# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: Adrian Sampson <adrian@radbox.org>

pkgname=python-musicbrainzngs
pkgver=0.7.1
pkgrel=6
pkgdesc="Python bindings for Musicbrainz' NGS webservice"
arch=(any)
url=https://github.com/alastair/python-musicbrainzngs
license=(BSD)
depends=(python)
makedepends=(
  git
  python-setuptools
)
source=(git+https://github.com/alastair/python-musicbrainzngs.git#tag=e4bca39720c7c0a1de38b1663faa3475cf5b7eff)
sha256sums=(SKIP)

pkgver() {
  cd python-musicbrainzngs

  git describe --tags | sed 's/^v//'
}

build() {
  cd python-musicbrainzngs

  python setup.py build
}

package() {
  cd python-musicbrainzngs

  python setup.py install --root="${pkgdir}" --optimize=1 --skip-build
  install -Dm 644 COPYING -t "${pkgdir}"/usr/share/licenses/python-musicbrainzngs/
}

# vim: ts=2 sw=2 et:
